Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 22
Line Integrals

Initializations

```
> restart;
  with(VectorCalculus):
  BasisFormat(false):
  with(plots):
```

22.1 Line Integrals with respect to Arc Length

Examples

Example 22.1.1

Let *C* denote the curve

$$\mathbf{r}(t) = \langle t+1, t^2 - t + 1\rangle, \ -2 \le t \le 3$$

Compute

$$\int_C x^2 y + 50\, ds$$

Solution
Code the integrant and the curve *C*.

```
> f:=(x,y)->x^2*y+50;
  r:=<t+1, t^2-t+1>;
```

$$f := (x, y) \to x^2 y + 50$$

$$r := \begin{bmatrix} t+1 \\ t^2 - t + 1 \end{bmatrix} \qquad (2.1.1.1)$$

Code the desired integral.

```
> e1:=Int(f(r[1], r[2])*sqrt(diff(r[1], t)^2+diff(r[2], t)
  ^2), t=-2..3);
```

$$e1 := \int_{-2}^{3} \left((t+1)^2 (t^2 - t + 1) + 50 \right) \sqrt{2 + 4t^2 - 4t}\, dt \qquad (2.1.1.2)$$

Of course this integral can be evaluated using an appropriate trigonometric substitution. You know all about that, right?

```
> answer:=value(e1);
  evalf(%);
```

$$answer := \frac{142835}{768}\sqrt{26} - \frac{6593}{256}\ln\left(-5+\sqrt{26}\right)$$

$$1007.885638 \qquad (2.1.1.3)$$

In the picture below this number is represented as the area of the region on the green parabolic cylinder above the xy plane and below the blue surface $z = f(x, y)$.

```
> p1:=plot3d([r[1], r[2], z], t=-2..3, z=0..f(r[1], r[2]),
  color=[0.2, 0.8, 0.1], style=patch, grid=[25, 10]):
  p2:=spacecurve([r[1], r[2], f(r[1], r[2])], t=-2..3,
  color=red, thickness=4):
  p3:=plot3d(f(x, y), x=-1..4, y=0..7, color=blue):
  display([p1, p2, p3], orientation=[-144, 60], axes=frame,
  labels=[x, y, z], style=wireframe);
```



22.2 Line Integrals with respect to variables other than Arc Length

Examples

Example 22.2.1

Let *C* denote the curve

$$\mathbf{r}(t) = \langle \sin t, t^2 \rangle, \ 0 \le t \le \pi$$

Compute

$$\int_C x^2 + y^3 \, dx$$

Solution

Code the integrant and the curve *C*.

```
> f:=(x, y)->x^2+y^3;
```

```
r:=<sin(t), t^2>;
```

$$f := (x, y) \to x^2 + y^3$$

$$r := \begin{bmatrix} \sin(t) \\ t^2 \end{bmatrix} \quad \textbf{(3.1.1.1)}$$

Create and evaluate the desired integral.

```
> e1:=Int(f(r[1], r[2])*diff(r[1], t), t=0..Pi);
```

$$e1 := \int_0^{\pi} \left(\sin(t)^2 + t^6\right) \cos(t)\, dt \quad \textbf{(3.1.1.2)}$$

```
> answer:=value(e1);
  evalf(%);
```

$$answer := -6\pi^5 + 120\pi^3 - 720\pi$$

$$-377.311618 \quad \textbf{(3.1.1.3)}$$

Example 22.2.2

Let C denote the closed curve bounded by

$$y = x^2 \quad \text{and} \quad y = 9$$

oriented counter clockwise wise. Compute

$$\int_C 5xy^2\, dx + x\sqrt{y}\, dy$$

Solution

Let $f(x, y) = 5xy^2$, $g(x, y) = x\sqrt{y}$ and parametrize the two pieces of the closed curve C..

```
> f:=(x, y)->5*x*y^2;
  g:=(x,y)->x*sqrt(y);
  r1:=<t,9>;
  r2:=<t,t^2>;
```

$$f := (x, y) \to 5\, x\, y^2$$

$$g := (x, y) \to x\sqrt{y}$$

$$r1 := \begin{bmatrix} t \\ 9 \end{bmatrix}$$

$$r2 := \begin{bmatrix} t \\ t^2 \end{bmatrix} \quad \textbf{(3.1.2.1)}$$

Plot the curve C.

```
> plot([[r1[1], r1[2], t=3..-3], [r2[1], r2[2], t=-3..3]],
  color=sienna);
```




```
> e1:=Int(f(r1[1], r1[2])*diff(r1[1], t), t=3..-3)+Int(g(r2
[1], r2[2])*diff(r2[2], t), t=-3..3);
```

$$e1 := \int_{3}^{-3} 405\, t \, dt + \int_{-3}^{3} 2\, t^2 \sqrt{t^2}\, dt \qquad \textbf{(3.1.2.2)}$$

```
> answer:=value(e1);
```

$$answer := 81 \qquad \textbf{(3.1.2.3)}$$

Observe that even though we start and end at the same point, the integral is **not** zero. In general, line integrals depend not only on the beginning and end point, but also on the path in between.

Example 22.2.3

Find the work W performed by the force field

$$\mathbf{F} = \langle x^2 + 30, -2\, xy - x^3 + 2\, x^2 \rangle$$

when a particle moves one time along the top half of the ellipse

$$9\, x^2 + 4\, y^2 = 36$$

in clockwise direction.

Solution

The indicated trajectory C can be parametrized as

$$\mathbf{r}(t) = \langle 2 \cos t, -3 \sin t \rangle, \quad -\pi \le t \le 0$$

Display the trajectory and the force field in one picture.

```
> F:=(x,y)-><x^2+30, -2*x*y-x^3+2*x^2>:
F(x,y);
r:=<2*cos(t), -3*sin(t)>;
```

$$\begin{bmatrix} x^2 + 30 \\ -2\, x\, y - x^3 + 2\, x^2 \end{bmatrix}$$

$$r := \begin{bmatrix} 2 \cos(t) \\ -3 \sin(t) \end{bmatrix} \qquad \textbf{(3.1.3.1)}$$

```
> p1:=fieldplot(F(x, y), x=-3..3, y=0..3.5, grid=[10, 10],
  arrows=SLIM, color=magenta):
  p2:=plot([r[1], r[2], t=-Pi..0], color=gold, thickness=3)
  :
  display([p1, p2], scaling=constrained, axes=frame);
```



The work done by the force field is given by

$$F = \int_C \mathbf{F} \cdot d\mathbf{r} = \int_{-\pi}^{0} \mathbf{F}(\mathbf{r}(t)) \cdot \mathbf{r}'(t)\, dt$$

Maple makes quick work of this integral.

```
> e1:=Int(F(r[1], r[2]).diff(r, t), t=-Pi..0);
```

$$e1 := \int_{-\pi}^{0} \left(-2\left(4\cos(t)^2 + 30\right)\sin(t) - 3\left(12\cos(t)\sin(t) - 8\cos(t)^3 + 8\cos(t)^2\right)\cos(t)\right) dt \quad \textbf{(3.1.3.2)}$$

```
> W:=value(e1);
  evalf(%);
```

$$W := \frac{448}{3} + 9\pi$$

$$177.6076672 \quad \textbf{(3.1.3.3)}$$